
https://lastminuteengineers.com/wemos-d1-mini-pinout-reference/

Omissa kokeissa kannattaa aluksi käyttää yläkuvassa esiintyviä vihreitä pinnejä. Muilla pinneillä on varattuja toimintoja, ja niihin kytketyt laitteet saattavat häiritä prosessorin normaalia toimintaa.

Tässä lyhyessä ohjeessa käydään läpi muutamia esimerkkejä, kuinka käyttää langatonta WiFi – verkkoa, ja kuinka kommunikoida sen yli.

On hyvä huomata, että ESP8266 – moduulille on useita erilaisia kirjastoja, esimerkiksi HTTP – serverin ylläpitoon, ja TCP/IP / UDP – kommunikoiniin, jotka poikeavat hieman toisistaan. 

Jotkut netistä löytyvistä esimerkeistä tarvitsevat siis lisäkirjastoja, jotka täytyy asentaa erikseen.

Samoin palvelimen käyttö saattaa vaatia kokonaan uusien toimintojen asennuksen, kuten tiedostoärjestelmän (SPIFFS). Nämä seikat ovat tämän perusohjeen aihepiirin ulkopuolella.



Wifi – verkko

Langattoman verkoon liittyy muutama seikka.

1) Liitytäänkö olemassaolevaan verkkoon, eli vaikkapa kotiverkkoon, joka on jo yhteydessä internettiin. Tällöin liittymän piste (Acces Point eli AP) on esimerkiksi oma kotinetti. Jotta voit liittyä olemassaolevaan verkkoon, tarvitset luonnollisesti salasanan ja verkon nimen (SSID). Jos liityt verkkoon, joka on liitetty internettiin, voit lähettää/vastaanottaa viestejä laitteiden välillä, ympäri maailmaa.

2) Voit määritellä laitteelle useita liityntäpisteitä, salasanoineen, ja laite liittää sinut aina “vahvimpaan” verkkoon.

3) Jos taas haluat luoda oman verkon, verkolle täytyy valita, nimi ja salasana. Tällöin oma laitteesi toimii AP-moodissa. Jos luot oman sisäverkon (local), voi halutessasi antaa omalle leitteellesi nimen ().

Yhdistetään olemassaolevaan verkkoon:

#include <ESP8266WiFi.h>        // Wifi kirjasto
const char* ssid     = "SSID";         // Verkon AP: nimi
const char* password = "PASSWORD";     // Salasana, jos sitä ei ole pelkät “”

void setup() {
  Serial.begin(115200);         // 
  delay(10);
  Serial.println('\n');

  WiFi.begin(ssid, password);             // Laite aloittaa pyrkii liittymään verkkoon
  Serial.print("Connecting to ");
  Serial.print(ssid); 
Serial.println(" ...");

  int i = 0;
  while (WiFi.status() != WL_CONNECTED) { // Jos laite ei ole yhdistynyt
    delay(1000);
    Serial.print(++i); Serial.print(' ');
  }

  Serial.println('\n');
  Serial.println("Connection established!");  
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());         // Tätä tarvitset kun olet yhteydessä laitteeseen
void loop() { }


Yhdistetään voimakkaimpaan verkkoon:

#include <ESP8266WiFi.h>        // Wifi kirjasto
#include <ESP8266WiFiMulti.h>   // Mahdollistaa useampaan verkoon liittymisen

ESP8266WiFiMulti wifiMulti;     // ESP8266WiFiMulti class, called 'wifiMulti'

void setup() {
  Serial.begin(115200);         // Sarjaliikenne nopeus 115200
  delay(10);
  Serial.println('\n');

  wifiMulti.addAP("ssid_AP_1", "your_passwordAP_1");   //  Verkon nimi ja salasana
  wifiMulti.addAP("ssid_AP_2", "your_password_AP_2");
  wifiMulti.addAP("ssid_AP_3", "your_password_AP_3");

  Serial.println("Connecting ...");
  int i = 0;
  while (wifiMulti.run() != WL_CONNECTED) { // Liitytään vahvimpaan verkkoon
    delay(1000);
    Serial.print('.');
  }
  Serial.println('\n');
  Serial.print("Connected to ");
  Serial.println(WiFi.SSID());              // Verkko johon yhdistyttiin
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());           // Laitteen IP 
}

. 

Normaalisti laitteeseen viitataan sen IP:n kautta. Joskus IP:n käyttö on hankalaa, ja siinä tapauksessa laitteelle voidaan antaa myös, netistä tuttu, sekokielinen nimi. Esimerkki tapauksessa esp8266.local.

Tässä tapauksessa siihen voidaan viitata, esimerkiksi selaimella, tyyliin laitteen_nimi.local.

#include <ESP8266WiFi.h>        // 
#include <ESP8266WiFiMulti.h>   // Mahdollistaa useita liittymispisteitä
#include <ESP8266mDNS.h>        // Paikallinen nimipalvelu

ESP8266WiFiMulti wifiMulti;     // Useita mahdollisia liittymispisteitä

void setup() {
  Serial.begin(115200);         // Start the Serial communication to send messages to the computer
  delay(10);
  Serial.println('\n');

  wifiMulti.addAP("ssid_AP_1", "your_passwordAP_1");   //  Verkon nimi ja salasana
  wifiMulti.addAP("ssid_AP_2", "your_password_AP_2");
  wifiMulti.addAP("ssid_AP_3", "your_password_AP_3");

  Serial.println("Connecting ...");
  int i = 0;
  while (wifiMulti.run() != WL_CONNECTED) { // 
    delay(1000);
    Serial.print(++i); Serial.print(' ');
  }
  Serial.println('\n');
  Serial.print("Connected to ");
  Serial.println(WiFi.SSID());              // mihin verkkoon olemme liittyneet
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());           // lähiverkon IP

  if (!MDNS.begin("esp8266")) {             // Antaa tälle laitteelle nimen esp8266.local
    Serial.println("Error setting up MDNS responder!");
  }
  Serial.println("mDNS responder started");
}

void loop() { }


AccesPoint – moodi. Jos haluat keskustella ESP:n kanssa suoraan, voit asettaa laitteen itsessään AP- moodiin. Tällöin otat yhteyden suoraan laitteen kanssa, mutta et ole enää yhetydessä Internettiin. 

#include <ESP8266WiFi.h>        //Wifi kirjasto

const char *ssid = "ESP8266 Access Point"; //  Verkon nimi, ei laitteen
const char *password = "";   // Kokeilua varten ei tarvita salasanaa

void setup() {
  Serial.begin(115200);
  delay(10);
  Serial.println('\n');

  WiFi.softAP(ssid, password);             // Aloitetaan AP - moodissa
  Serial.print("Access Point \"");
  Serial.print(ssid);
  Serial.println("\" started");

  Serial.print("IP address:\t");
  Serial.println(WiFi.softAPIP());         // IP - osoite
}

void loop() { }

Laite on vain yhdessä verkossa kerrallaan. 
Yhteyden testaaminen:
Liity verkkoon: ESP8266 Acces point
Windosissa avaa komentoikkuna: "Command Prompt".
Macissa: "Terminal".
Kirjoita terminaaliin ping 192.168.4.1

Yhteys toimii.



HTTP – palvelin.

Yleinen tapa käyttää esp8266 on tehdä sitä palvelin. Käytännössä alkuun pääseen kahdella erilaisella komennolla GET ja POST. Get käskyllä pyydetään palvelimelta tietoa, lähinnä html-sinvua, ja POST käskyllä lähetetään palvelimelle tietoa.
https://www.martyncurrey.com/esp8266-and-the-arduino-ide-part-2-control-an-led-from-a-web-page-using-access-point-mode-ap/

#include <ESP8266WiFi.h>
#include <WiFiClient.h>
#include <ESP8266WiFiMulti.h> 
#include <ESP8266WebServer.h>   // Include the WebServer library

ESP8266WiFiMulti wifiMulti;     // Create an instance of the ESP8266WiFiMulti class, called 'wifiMulti'
ESP8266WebServer server(80);    // Create a webserver object that listens for HTTP request on port 80
void setup(void){
  Serial.begin(115200);         // Start the Serial communication to send messages to the computer
  delay(10);
  Serial.println('\n');
  wifiMulti.addAP("Tmu", "Porkkana1");   // add Wi-Fi networks you want to connect to
  wifiMulti.addAP("ssid_from_AP_2", "your_password_for_AP_2");
  Serial.println("Connecting ...");
  while (wifiMulti.run() != WL_CONNECTED) { // Wait for the Wi-Fi to connect: 
    delay(250);
    Serial.print('.');
  }
  Serial.println('\n');
  Serial.print("Connected to ");
  Serial.println(WiFi.SSID());              // Tell us what network we're connected to
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());           // 
  server.on("/", handleRoot);               // 
  server.onNotFound(handleNotFound);        // 
  server.begin();                           // Actually start the server
  Serial.println("HTTP server started");
}
void loop(void){
  server.handleClient();                    // Listen for HTTP requests from clients
}
void handleRoot() {
  server.send(200, "text/plain", "Hello world!");   // Send HTTP status 200 (Ok) 
}
void handleNotFound(){
  server.send(404, "text/plain", "404: Not found"); // Send HTTP status 404 (Not Found) }
/*
 * Intended to be run on an ESP8266 d1 clone
 * use web broswer to go to 192.168.4.1
*/
#include <ESP8266WiFi.h>
const char AP_NameChar[] = "LEDControl" ;
const char WiFiPassword[] = "";
WiFiServer server(80);
String header = "HTTP/1.1 200 OK\r\nContent-Type: text/html\r\n\r\n";
String html_1 = "<!DOCTYPE html><html><head><title>LED Control</title></head><body><div id='main'><h2>LED Control</h2>";
String html_2 = "<form id='F1' action='LEDON'><input class='button' type='submit' value='LED ON' ></form><br>";
String html_3 = "<form id='F2' action='LEDOFF'><input class='button' type='submit' value='LED OFF' ></form><br>";
String html_4 = "</div></body></html>";
String request = "";
int LED_Pin = 2;
void setup() 
{
Serial.begin(115200);
    pinMode(LED_Pin, OUTPUT); 
    boolean conn = WiFi.softAP(AP_NameChar, WiFiPassword);
    server.begin();
} // void setup()
void loop() 
{
    // Check if a client has connected
    WiFiClient client = server.available();
    if (!client)  {  return;  }
    // Read the first line of the request
    request = client.readStringUntil('\r');
 Serial.println(request);
    if       ( request.indexOf("LEDON") > 0 )  { digitalWrite(LED_Pin, LOW);  }
    else if  ( request.indexOf("LEDOFF") > 0 ) { digitalWrite(LED_Pin, HIGH);   }
    client.flush();
 
    client.print( header );
    client.print( html_1 );
    client.print( html_2 );
    client.print( html_3 );
    client.print( html_4);
    delay(5);
} // void loop()



/* IP address: 192.168.4.1 
Testaa http://192.168.4.1/h
Testaa http://192.168.4.1/t
*/

#include "Arduino.h"
#include "ESP8266WiFi.h"
#include "ESPAsyncWebServer.h"
const char* ssid = "ESP8266";
const char* password = "";
String request = "";
float i=0;
float tem = 0.0;
float hem = 0.0;
AsyncWebServer server(80);
const char index_html[] PROGMEM = R"rawliteral(
ESP8266 Number Server <br>
Number tem
%t% Yksikko <br>
Number hem
%h% Yksikko <br>
)rawliteral"; // korvataan tem ja hem t:llä ja h:lla
String processor(const String& var){
Serial.println(var);
if(var == "t"){
return String(tem);
} else if(var == "h"){
return String(hem); }
return String(); }
void setup(){
Serial.begin(115200);// Sarjaliikenne debuggaukseen
Serial.println();
Serial.println("Setting AP (Access Point)…");
WiFi.softAP(ssid, password);
IPAddress IP = WiFi.softAPIP();
Serial.print("AP IP address: ");
Serial.println(IP);
server.on("/", HTTP_GET, [](AsyncWebServerRequest *request){  // palauttaa koko sivun
request->send_P(200, "text/html", index_html, processor); });
// Palauttaa vain muuttujan t arvon, muutettuna merkeiksi, http://192.168.4.1/t
server.on("/t", HTTP_GET, [](AsyncWebServerRequest *request){
request->send_P(200, "text/plain", String(tem).c_str()); }); // tem tyypiksi string
server.on("/h", HTTP_GET, [](AsyncWebServerRequest *request){ // vain muuttuja h
request->send_P(200, "text/plain", String(hem).c_str());  }); // hem tyypiksi string
server.begin();  }// Start server
void loop(){   tem=i; hem=i; i++; }


Automaattinen päivitys voidaa tehdä vaikkapa näin:

#include "Arduino.h"
#include "ESP8266WiFi.h"
#include "ESPAsyncWebServer.h"
const char* ssid = "ESP8266";
const char* password = "";
String request = "";
float i=0;
float tem = 0.0;
float hem = 0.0;
AsyncWebServer server(80);
const char index_html[] PROGMEM = R"rawliteral(
   <head>
  <meta http-equiv="refresh" content="1"> 
</head>
ESP8266 Number Server
Number tem
%t% Yksikko
)rawliteral"; // korvataan tem automaattisesti

String processor(const String& var){
Serial.println(var);
if(var == "t"){
return String(tem);
}
return String();
}
void setup(){
Serial.begin(115200);// Sarjaliikenne debuggaukseen
Serial.println();
Serial.println("Setting AP (Access Point)…");
WiFi.softAP(ssid, password);
IPAddress IP = WiFi.softAPIP();
Serial.print("AP IP address: ");
Serial.println(IP);
// Palauttaa kokonaisen sivun
server.on("/", HTTP_GET, [](AsyncWebServerRequest *request){ 
request->send_P(200, "text/html", index_html, processor);
});
server.begin(); // Start server
}
void loop(){ delay(1000);  tem=i++;
}
Lisää ohjeita: https://www.martyncurrey.com/esp8266-and-the-arduino-ide-part-8-auto-update-webpage/

GET/POST käskyt. Edellä on käytetty GET – käskyä. Tätä kytetän yleisesti, kun haetaan HTTP-sivua. Kun halutaan lähettää tietoa, vaikkapa teksitikenttien sisältöä, käytetyään yleneensä POST- käskyä. Se ei näy selaimen käskyikkunassa, mutta ei ole myöskään suojattu. Tämän lisäksi valikoimasta löytyy useita muitakin kommunikointikäskyjä.




Telnet TCP ja UDP

HTPP palvelin on vain yksi tata kommunikoida ESP:n kanssa. Usein on järkevämpää käyttää muita tapoja, joista tässä tarkastellaan vain muutamaan. Tähän kokeiluun tarvitse ensiksi telnet ohjelman (tai vastaavan).

Ohjeet telnetin asennukseen löydät:
https://www.layerstack.com/resources/tutorials/Installing-Telnet-Client-on-macOS
https://blog.mrkeyshop.com/en/guides/how-to-enable-and-use-telnet-on-windows-11-and-10/
https://www.layerstack.com/resources/tutorials/Installing-telnet-on-Linux-and-Windows-Cloud-Servers

Lataa seuraavan sivun koodi esp:lle. Käytä aluksi olemassaolevaa AP:tä, vaikkapa kotinetin tai koulun.

Avaa komentoikkuna, ja kirjoita  telnet 192.168.51.46 8888

➜  ~ telnet 192.168.51.46 8888
Trying 192.168.51.46...
Connected to 192.168.51.46.
Escape character is '^]'.
tama lahtee nyt telnetille Arduinosta
tama lahtee ESP82:lle Telnetista 

Avaa Arduinon sarjaliikenne, ja kirjoita sinne jokin viesti:

Näillä ohjelmilla pääsee tutustumaan TCP – pakettien lähettelyyn.




#include <ESP8266WiFi.h> // Yksinkertainen TCP/IP pakettien testaus
int port = 8888;  //Port number
WiFiServer server(port);
const char *ssid = "Tmu";  //Enter your wifi SSID
const char *password = "Porkkana1";  //Enter your wifi Password
int count=0;
void setup() 
{
  Serial.begin(115200);
  Serial.println();
  WiFi.mode(WIFI_STA);
  WiFi.begin(ssid, password); //Connect to wifi
  Serial.println("Connecting to Wifi"); // Wait for connection  
  while (WiFi.status() != WL_CONNECTED) {   
    delay(500);
    Serial.print(".");
    delay(500);
  }
  Serial.print("Connected to ");
  Serial.println(ssid);
  Serial.print("IP address: ");
  Serial.println(WiFi.localIP());  
  server.begin();
  Serial.print("Open Telnet and connect to IP:");
  Serial.print(WiFi.localIP());
  Serial.print(" on port ");
  Serial.println(port);
}
void loop() 
{
  WiFiClient client = server.available();
  if (client) {
    if(client.connected())
    {  Serial.println("Client Connected");   }
    while(client.connected()){      
      while(client.available()>0){
        // read data from the connected client
        Serial.write(client.read()); 
      }
      while(Serial.available()>0) //Send Data to connected client
      { client.write(Serial.read());}
    }
    client.stop();
    Serial.println("Client disconnected");    
  }
}
import processing.net.*;
Client c;
String data;

void setup() {
  size(200, 200);
  background(50);
  fill(200);
  c = new Client(this, "192.168.51.46", 8888);  // Connect to server on port 8888
  c.write("Tama lahti nyt Processingista");  // 
}

void draw() {
  if (c.available() > 0) {    // If there's incoming data from the client...
    data += c.readString();   // ...then grab it and print it
    println(data);
  }
}

Tässä yhteydessä jätetään UDP – pakettien lähettäminen pois. UDP – paketit lähetetään matkaan, mutta niiden perillemenoa ei tarkisteta.


Parseri

Usein viestit kannattaa lähettää yhtenä ‘pötkönä’ eroteltuna vaikkapa pilkulla (CSV – Comma Sparated Values). Tämä tarvitsee kuitenkin vastaanottopäässä hieman enemmän toimia, sillä arvot on eriteltävä ja muunnettava oikeisiin tyyppeihin. Viesti on aina char – tyyppiä.
import processing.net.*;
Client c;
String input;
int data[];
void setup() 
{
  size(450, 255);
  background(204);
  stroke(0);
  frameRate(5); 
  c = new Client(this, "192.168.51.46", 8888); //Oma IP ja portti
}

void draw() 
{
  if (mousePressed == true) {
    // Draw our line
    stroke(255);
    line(pmouseX, pmouseY, mouseX, mouseY);
    // Send mouse coords to other person
    c.write(pmouseX + " " + pmouseY + " " + mouseX + " " + mouseY + "\n");
  }
  // Receive data from server
  if (c.available() > 0) {
    input = c.readString();
    println(input);// tulostetaan saapunut raaka data
    input = input.substring(0, input.indexOf("\n")); // Only up to the newline
    data = int(split(input, ',')); // jaetaan pilkulla erotetut arvot
    stroke(0); // piirrellään jotta nähdään että yhteys toimii
    line(data[0], data[1], data[2], data[3]);
    println(data[0]+data[1]+data[2]+data[3]); // tarkisteaan tyyppi
  }
}    stroke(0);
    line(data[0], data[1], data[2], data[3]);
    println(data[0]);
  }
}




#include <ESP8266WiFi.h> // Yksinkertainen TCP/IP pakettien testaus
int port = 8888;  //Port number
WiFiServer server(port);
const char *ssid = "Tmu";  //Enter your wifi SSID
const char *password = "Porkkana1";  //Enter your wifi Password

void setup()  {
  Serial.begin(115200);
  Serial.println();
  WiFi.mode(WIFI_STA);
  WiFi.begin(ssid, password); //Connect to wifi
  Serial.println("Connecting to Wifi"); // Wait for connection  
  while (WiFi.status() != WL_CONNECTED) {   
    delay(500);
    Serial.print(".");
    delay(500);  }
  Serial.print("Connected to ");
  Serial.println(ssid);
  Serial.print("IP address: ");
  Serial.println(WiFi.localIP());  
  server.begin();
  Serial.print("Open Telnet and connect to IP:");
  Serial.print(WiFi.localIP());
  Serial.print(" on port ");
  Serial.println(port); }
void loop() 
{
  WiFiClient client = server.available();
  if (client) {
    if(client.connected())
    {  Serial.println("Client Connected");   }
    while(client.connected()){      
      String postStr = String(random(300));
           postStr +=",";
           postStr += String(random(300));
           postStr +=",";
           postStr += String(random(300));
           postStr +=",";
           postStr += String(random(300));
           postStr += "\n";
        client.write(postStr.c_str());delay(1000);
    }
    client.stop();
    Serial.println("Client disconnected");    
  }
}




Tiedon välitys parser

Yleisimmät tavat joilla tietoa välitetään ovat CSV (Comma Sparater Values) tai Json.
Ensimäinen on hyvä, jos tietoa halutaan käsitellä myöhemmin, vaikkapa Exeli:ssä. Json taas on hieman kehittyneempi, ja sen rakenne on yksinkertaisesti muuttuja, arvo.

Object
¨

Example: myJSON = {"name":"John", "age":31, "city":"New York"}

Example: myJSON = {"data": [100,20,34,120],"name":"Voltage"}

Example: myJSON = "{"Name":"circuits", "Data": 45, "DataSet":[3,4,56,2,2,56], "check":true}"

Json – tiedostojen avaamisessa on apua: https://jsonformatter.org/json-viewer .
Json – on yleisessä käytössä, ja sekä Arduino, että Processing tukevat tätä tiedostomuotoa.

//Json testi:
#include <ESP8266WiFi.h>
#include <ESP8266WebServer.h>
#include <ArduinoJson.h>
#define SSID "Tmu"
#define PASSWORD "Porkkana1"
String id = SSID;
String pass = PASSWORD ;
String postNumber = "";
ESP8266WebServer server(80);  
void sendData(){
    StaticJsonDocument<300> JSONData;
    JSONData["key"] = "Value"; // voit lisätä uusia muuttujapareja
    JSONData["number"] = String(random(100)); // kaikki oltava string tyyppiä
    char data[300]; // lähetettävä paketti on oltava tätä muotoa
    serializeJson(JSONData,data); 
  server.send(200,"application/json",data); // 192.168.51.46/
}
void setup() {
    Serial.begin(115200);
    WiFi.begin(SSID,PASSWORD);
    while(WiFi.status() != WL_CONNECTED){
      Serial.println("Connecting");
    delay(1000);   }
    Serial.println("Connected to");
    Serial.println(WiFi.localIP());
    delay(500);
    server.on("/test",HTTP_GET,sendData); // kirjoita selaimeen 192.168.51.46/test
    server.begin(); }
void loop() { server.handleClient();  }

Kuva: testaus selaimella ja processin ympäristössä.

import http.requests.*; //Testing json processing:
public void setup() {
  size(400, 400);
  GetRequest get = new GetRequest("http://192.168.51.46/test");
  get.send(); // d program will wait untill the request is completed
  println("response: " + get.getContent());
  JSONObject response = parseJSONObject(get.getContent());
  println("number: " + response.getString("number"));   }



#include <ESP8266WiFi.h>
#include <ESPAsyncTCP.h>
#include <ESPAsyncWebServer.h>
const char* ssid = "TMU";
const char* password = "Porkkana1";
const int output = 2;
String sliderValue = "0";
const char* PARAM_INPUT = "value";
AsyncWebServer server(80);
const char index_html[] PROGMEM = R"rawliteral(
<!DOCTYPE HTML><html>
<head>
  <meta name="viewport" content="width=device-width, initial-scale=1">
  <title>ESP Web Server / Slider </title>
  <style>
    html {font-family: Arial; display: inline-block; text-align: center;}
    h2 {font-size: 2.3rem;}
    p {font-size: 1.9rem;}
    body {max-width: 400px; margin:0px auto; padding-bottom: 25px;}
    .slider { -webkit-appearance: none; margin: 14px; width: 360px; height: 25px; background: #FFD65C;
      outline: none; -webkit-transition: .2s; transition: opacity .2s;}
    .slider::-webkit-slider-thumb {-webkit-appearance: none; appearance: none; width: 35px; height: 35px; background: #003249; cursor: pointer;}
    .slider::-moz-range-thumb { width: 35px; height: 35px; background: #003249; cursor: pointer; } 
  </style>
</head>
<body>
  <h2>ESP Web Server</h2>
  <p><span id="textSliderValue">%SLIDERVALUE%</span></p>
  <p><input type="range" onchange="updateSliderPWM(this)" id="pwmSlider" min="0" max="1023" value="%SLIDERVALUE%" step="1" class="slider"></p>
<script>
function updateSliderPWM(element) {
  var sliderValue = document.getElementById("pwmSlider").value;
  document.getElementById("textSliderValue").innerHTML = sliderValue;
  console.log(sliderValue);
  var xhr = new XMLHttpRequest();
  xhr.open("GET", "/slider?value="+sliderValue, true);
  xhr.send();
}
</script>
</body>
</html>
)rawliteral";
String processor(const String& var){
  Serial.println(var);
  if (var == "SLIDERVALUE"){
    return sliderValue;
  }
  return String();
}
void setup(){
  Serial.begin(115200);
  analogWrite(output, sliderValue.toInt());// sliderValue on merkkijono
  WiFi.begin(ssid, password);
  while (WiFi.status() != WL_CONNECTED) {
    delay(1000);
    Serial.println("Connecting to WiFi..");
  }
  Serial.println(WiFi.localIP());
  server.on("/", HTTP_GET, [](AsyncWebServerRequest *request){
    request->send_P(200, "text/html", index_html, processor);
  });

  //  GET request <ESP_IP>/slider?value=<inputMessage>
  server.on("/slider", HTTP_GET, [] (AsyncWebServerRequest *request) {
    String inputMessage;
    // GET input1 value on <ESP_IP>/slider?value=<inputMessage>
    if (request->hasParam(PARAM_INPUT)) {
      inputMessage = request->getParam(PARAM_INPUT)->value();
      sliderValue = inputMessage;
      analogWrite(output, sliderValue.toInt());
    }
    else {
      inputMessage = "No message sent";
    }
    Serial.println(inputMessage);
    request->send(200, "text/plain", "OK");
  });
  server.begin();
}
void loop() {  }

Ohessa pieni moottoriohjauskoodi esp8266 robottialustalle. Kesken.
#include <ESP8266WiFi.h>
#include <Servo.h>

// Pin definitions for servos
#define servoPin1 5 // Servo 1 for forward/backward
#define servoPin2 14 // Servo 2 for left/right

// Speed Control: Range from 0 (slow) to 180 (fast)
int speed = 90; // Default speed

// Wi-Fi Access Point credentials
const char* ssid = "ESP8266_Car1";
const char* password = "";

// Web server object
WiFiServer server(80);

// Servo objects
Servo servo1;
Servo servo2;

void setup() {
  Serial.begin(115200);

  // Initialize the servos
  servo1.attach(servoPin1);
  servo2.attach(servoPin2);

  // Start Wi-Fi in AP mode
  WiFi.softAP(ssid, password);
  Serial.println("Access Point Started");
  Serial.print("IP Address: ");
  Serial.println(WiFi.softAPIP());

  // Start the web server
  server.begin();
}

void loop() {
  WiFiClient client = server.available(); // Check if there's a client

  if (client) {
    String currentLine = "";

    // Handle client requests
    while (client.connected()) {
      if (client.available()) {
        char c = client.read();
        Serial.write(c); // For debugging

        if (c == '\n') {
          // If the line is empty, it's the end of the HTTP request
          if (currentLine.length() == 0) {
            // Send the HTTP response header
            client.println("HTTP/1.1 200 OK");
            client.println("Content-Type: text/html");
            client.println("Connection: close");
            client.println();

            // Serve HTML with buttons and speed slider
            client.println("<html>");
            client.println("<head><title>Robot Controller</title></head>");
            client.println("<body>");
            client.println("<h1>Control Your Car</h1>");
            
            // Buttons for movement control
            client.println("<form action=\"/forward\" method=\"get\"><button type=\"submit\">-------    Forward    -------</button></form>");
            client.println("<form action=\"/left\" method=\"get\"><button type=\"submit\">Left</button>");
             client.println("<form action=\"/stop\" method=\"get\"><button type=\"submit\">Stop</button>");
            client.println("<form action=\"/right\" method=\"get\"><button type=\"submit\">Right</button></form>");
            client.println("<form action=\"/backward\" method=\"get\"><button type=\"submit\">------   Backward   ------</button></form>");
            
            // Speed control slider
            client.println("<h3>Speed Control</h3>");
            client.println("<form action=\"/speed\" method=\"get\">");
            client.println("<input type=\"range\" name=\"speed\" min=\"0\" max=\"180\" value=\"" + String(speed) + "\" onchange=\"this.form.submit()\">");
            client.println("<p>Speed: <span id=\"speedValue\">" + String(speed) + "</span></p>");
            client.println("</form>");

            client.println("<script>document.querySelector('input[type=range]').addEventListener('input', function() { document.getElementById('speedValue').textContent = this.value; });</script>");
            client.println("</body>");
            client.println("</html>");

            break;
          } else {
            // Control movement based on button press
            if (currentLine.indexOf("GET /forward") >= 0) {
              moveForward();
            } else if (currentLine.indexOf("GET /backward") >= 0) {
              moveBackward();
            } else if (currentLine.indexOf("GET /left") >= 0) {
              turnLeft();
            } else if (currentLine.indexOf("GET /right") >= 0) {
              turnRight();
            } else if (currentLine.indexOf("GET /stop") >= 0) {
              stopMovement();
            } else if (currentLine.indexOf("GET /speed") >= 0) {
              setSpeed(currentLine);
            }
          }
          currentLine = "";
        } else if (c != '\r') {
          currentLine += c; // Accumulate the HTTP request
        }
      }
    }
    client.stop();
  }
}

// Function to move the car forward
void moveForward() {
  servo1.write(90 - speed);  // Forward movement for servo 1
  servo2.write(90 + speed);  // Forward movement for servo 2
}

// Function to move the car backward
void moveBackward() {
  servo1.write(90 + speed);  // Backward movement for servo 1
  servo2.write(90 - speed);  // Backward movement for servo 2
}

// Function to turn the car left
void turnLeft() {
  servo1.write(90);          // Keep servo 1 neutral
  servo2.write(0);           // Turn servo 2 left
}

// Function to turn the car right
void turnRight() {
  servo1.write(90);          // Keep servo 1 neutral
  servo2.write(180);         // Turn servo 2 right
}

// Function to stop the car
void stopMovement() {
  servo1.write(90);  // Neutral position for servo 1
  servo2.write(90);  // Neutral position for servo 2
}

// Function to set the speed based on the slider value
void setSpeed(String currentLine) {
  int startIndex = currentLine.indexOf("speed=") + 6;
  int endIndex = currentLine.indexOf(" ", startIndex);
  if (endIndex == -1) {
    endIndex = currentLine.length();
  }
  speed = currentLine.substring(startIndex, endIndex).toInt();
  Serial.print("Speed: ");
  Serial.println(speed);
}




Hyviä linkkejä:
(Seuraavassa syytä kiinnittää huomiota ontureihin onko paine anturi juuri sama ja onko kosteus lämpötila-anturi dht11?)
https://raw.githubusercontent.com/RuiSantosdotme/ESP32-Course/master/code/WiFi_Web_Server_DHT/WiFi_Web_Server_DHT.ino

https://randomnerdtutorials.com/esp32-esp8266-web-server-outputs-momentary-switch/




Edistyneemmille:
Näissä on hyvä tutustua ainakin pintapuolisesti html ja JavaSkript sekä CSS -tekniikkaan: https://www.w3schools.com/html/default.asp )

https://gndtovcc.home.blog/2020/04/13/esp8266-web-server-using-spiffs-spi-flash-file-system-nodemcu/

https://randomnerdtutorials.com/esp8266-nodemcu-http-get-post-arduino/

